

**O‘zbekiston Respublikasi
Oliy va o‘rta maxsus ta‘lim vazirligi**

Namangan muhandislik – pedagogika instituti

“Informatika va elektroenergetika” fakulteti

**“Informatika va AT ” kafedrası
“Informatika va axborotlar texnologiyasi”
fanidan**

**Muammoli ma‘ruzalar matni
(2-qism)**



Namangan – 2005 y

«Informatika va axborotlar texnologiyasi» fanidan tayyorlangan ushbu muammoli ma`ruzalar matni 5140900-Kasb ta`limi «Informatika va axborotlar texnologiyasi» ta`lim yo`nalishi bo`yicha ta`lim olayotgan kunduzgi bo`lim talabalariga mo`ljallangan bo`lib, NamMPI ilmiy kengashi tomonidan tasdiqlangan na`munaviy o`quv dasturi asosida tayyorlandi.

Mualliflar: “Informatika va AT” kafedrası
katta o`qituvchisi A. Jumaboev
“Informatika va AT” kafedrası
katta o`qituvchisi N.Qurbonov

Taqrizchilar: “Xisoblash texnikasi” kafedrası dotsenti,
Yo.Tillaboyev
NamDU, “Informatika va amaliy matematika”
kafedrası mudiri, dotsent A.Imomov

Ushbu ma`ruzalar matni «Informatika va axborotlar texnologiyasi» kafedrasining ilmiy uslubiy seminarida ko`rib chiqilgan va institut ilmiy-uslubiy kengashiga ko`rib chiqish uchun tavsiya etilgan.
(1 sonli majlis bayoni 27 avgust 2005 yil).

Namangan muxandislik-pedagogika instituti ilmiy-uslubiy kengashi tomonidan qo`rib chiqilgan va chop etishga tavsiya etilgan
(1 sonli majlis bayoni 29 avgust 2005 yil).

So‘z boshi

Ishlab chiqarish jarayonlarini avtomatlashtirish, yangi texnologiyalarni loyihalash, boshqaruv va mehnat samaradorligini oshirish, uzluksiz ta‘lim tizimini rivojlantirish va sifatli kadrlar tayyorlashni tezlashtirish va yo‘nalishlarning barchasini hisoblash texnikasini rivojlantirishga va u esa, bugungi kunda yosh avlodni qanday saviyada tayyorlashga bog‘liq.

Davlatimiz tomonidan keyingi paytlarda ta‘lim tizimiga, ayniqsa, kollej va institutda “Informatika va axborotlar texnologiyasi” fanini o‘qitilishiga ahamiyat berilishi, zamonaviy kompyuterlar bilan ularni jixozlanishi o‘qitish tizimida keskin o‘zgarish bo‘lishiga sabab bo‘ldi. Xozirgi bitiruvchi yosh mutaxassislar ishlab chiqarishda to‘g‘ridan-to‘g‘ri aniq vazifalarni kompyuterda bajara olishi bunga dalil bo‘ladi.

Xuddi shunday maqsadlarda yozilgan ma‘ruzalar matnida shaxsiy kompyuterning dasturiy ta‘minotlari, operasion tizimlar, operasion qobig‘lar, WINDOWS operasion tizimi va matnli muxarrir, elektron jadval, xamda grafik muxarrirlar, ma‘lumotlar bazasini tashkil etish va boshqarishda qo‘llaniladigan dasturiy vositalar to‘g‘risida, shu bilan birgalikda kompyuter tarmog‘i, Internet xalkaro axborot tarmog‘i xaqida ma‘lumotlar berilgan.

Ma‘ruza matnlari kafedra o‘qituvchilarining uzoq yillab to‘plagan o‘qitish tajribalari, zamonaviy adabiyotlar, jurnallarda chop etilgan maqolalarga asoslanib yozilgan.

Mualliflar.

1-Ma'ruza. Takrorlash operatori va ularning qo'llanilishi. Takrorlanuvchi algoritmlarni dasturlash. Takrorlanuvchi jarayonlarni hisoblash algoritmlari va dasturlarini tuzish.

Reja:

1. Takrorlash operatori va ularning qo'llanilishi.
2. Takrorlanuvchi jarayonlar uchun algoritm blok-sxemalar tuzish.
3. Takrorlanuvchi jarayonlar uchun misollar ko'rish.

Agar algoritmning ma'lum bir qismi bir yoki bir necha bor takrorlansa, bunday algoritmlar **takrorlanuvchi algoritmlar** deb ataladi. Takrorlanuvchi jarayonlarning algoritmlarini biz oldingi ma'ruzalarimizda shartli buyruqlar yordamida yozdik. Shartli buyruqlarning qulayligi shundaki, uning yordamida takrorlanishlar soni oldindan ma'lum bo'lgan yoki ma'lum bo'lmagan takrorlanuvchi jarayonlarni ham ifodalash mumkin. Lekin bu jarayonni tashkil va nazorat qiluvchi biror bir o'zgaruvchiga (takrorlanuvchi jarayonning parametriga) boshlang'ich berishni, uning qiymatlarini o'zgartirib turishni va bu qiymatni bitta shartli buyruq amalga oshiradi. Shu shartni hammasini bitta takrorlash buyrug'i yordamida amalga oshirish mumkin va uning Beysik tilidagi ko'rinishi quyidagicha bo'ladi:

<N> FOR X=A TO B STEP H

[Takrorlanuvchi buyruqlar]

<M> NEXT [X]

Bu yerda

<N> va <M> - satr nomerlari;

FOR-buyruqning nomi bo'lib, «uchun» degan ma'noni bildiradi;

TO-yordamchi so'z bo'lib, «gacha» degan ma'noni bildiradi;

STEP-yordamchi so'z bo'lib, «qadam» degan ma'noni bildiradi;

X-butun yoki haqiqiy turdagi o'zgaruvchi bo'lib, takrorlanuvchi jarayon (sikl) ning parametri deb ataladi;

A, V va N-mos ravishda jarayon parametrining boshlang'ich, oxirgi qiymati va o'zgarish qadami bo'lib, umuman ifodalar bo'lishi mumkin. Jarayon boshlanishidan oldin bu ifodalarning qiymati bir marta hisoblanadi va oxirigacha o'zgarmaydi;

N - yordamchi so'z bo'lib, takrorlanuvchi jarayonning oxirini ko'rsatuvchi buyruq.

Agar parametrning qadami N birga teng bo'lsa, takrorlanuvchi jarayonning sarlavhasini quyidagicha ham yozish mumkin:

<N> FOR X=A TO B

Misol:

70 FOR I=1 TO 20 STEP 3

Bu misolda X o`zgaruvchi o`rnida haqiqiy turli I o`zgaruvchi, uning boshlang`ich qiymati A sifatida 1 soni, yuqori chegarasi V sifatida 20, o`zgarish qadami N sifatida 3 soni olingan. Bu misolda I o`zgaruvchi quyidagi qiymatlarni qabul qilishi mumkin:


1, 4, 7, 10, 13, 16, 19

Bundan ko`rinib turibdiki, I o`zgaruvchi yuqori chegara 20 ni qabul qilmasligi ham mumkin ekan. Sikldan chiqishda I ning qiymati 22 ga teng bo`ladi.

Quyidagi

70 FOR X=2 TO 10

misolda o`zgaruvchi 2 dan boshlab 10 gacha (10 ham kiradi) bir qadamlab qiymatlar qabul qiladi. Umuman takrorlash buyrug`ining ishlash tartibini quyida ko`rsatilgan 1.1 (a)-rasmidagi kabi blok-sxemada tasvirlash mumkin. Bu blok sxemadan ko`rinib turibdiki, takrorlash buyrug`i kamida bir marta bajariladi. Shartni tekshirishda SIGN funksiyasining olinishi qadamning manfiy qiymatlarini ham olishda kerak bo`ladi.



1.1 - rasm

Takrorlanuvchi buyruqlar ichida ham takrorlash buyruqlari bo`lishi mumkin. Bunday jarayonlar ichma-ich joylashgan takrorlanuvchi jarayonlar deb ataladi.

Ichma-ich joylashgan takrorlanuvchi jarayonlarning ko`p bo`lishi mashina vaqtini ko`p oladi. Masalan, 10 ta ichma-ich joylashgan takrorlash buyrug`ining har biri 10 martadan ishlasa, takrorlanuvchi buyruqlar jami 10^{10} marta bajariladi.

Takrorlanuvchi jarayonlarga misolar ko`rib o`tamiz:


1-misol. 1 dan 100 gacha bo`lgan toq sonlarning yig`indisini topish algoritmining blok-sxemasini (1.1(B)-rasm) va dasturini tuzamiz:

```

10 REM "YIGINDI"
20 S=0
30 FOR I=1 TO 100 STEP 2
40 S=S+I
50 NEXT I
60 ? S
70 END

```

2-misol. $\sin x$ va $\cos x$ funksiyalarining qiymatlarini x argumentning 0 dan 2π oraliqdagi qiymatlarida $0,1$ qadam bilan hisoblash:



```

10 REM "SINX BA COSX"
20 FOR X=0 TO 6.28 STEP 0.1
30 Y=SIN(X)
40 Z=COS(X)
50 PRINT X, Y, Z
60 NEXT
70 END

```


1.2-rasm

Nazariy savollar.

1. Takrorlash operatorining buyruq formatini yozib bering.
2. Takrorlanuvchi jarayonlarga misollar ko`rsating.
3. 1 dan 50 gacha bo`lgan juft sonlarning yig`indisini topish algoritmining blok-sxemasini va dasturini tuzing.
4. Ichma-ich joylashgan takrorlanuvchi jarayonlarga misol ko`rsating.

Muammoli savollar

1. $\langle N \rangle$ FOR $X = A$ TO B STEP H –ifodani to'liq mazmunini ochib bering.
- 2.



Chizmani to'ldiring.

3. 50 ta berilgan sonlar ichidan 3 ga karrali sonlarni ajratib olish dasturini tuzing.
4. Ichma-ich joylashgan takrorlanuvchi jarayonlarga misollar ko'rsating.

2-Ma'ruza. Jadval kattaliklar. Massiv. Massivlarni ifodalash va ular ustida amallar bajarish. Jadval kattaliklarini qayta ishlash.

Reja:

1. Jadval kattaliklar haqida umumiy tushunchalar.
2. Jadval kattaliklar ustida misollar bajarish.

Biz oldingi ma'ruzalarimizda o'zgaruvchi kattaliklar, ularning turlari, nomlari, ularning qiymatlarini kiritish va chiqarish bilan tanishgan edik. O'zgaruvchi kattaliklarning soni bitta yoki ikkita yoki uchta bo'lganda, ularni nomlash va ularga qiymat berish unchalik qiyinchilik tug'dirmaydi. Ammo ular 100 ta, 500 ta, 1000 ta bo'lsa-chi, ayrim hollarda bundan ham ko'p bo'lishi mumkin. Matematikada bunday hollarda vektorlar yoki matrisalardan foydalanilgani kabi dasturlashda ham bir yoki bir necha o'lchovli jadval kattaliklardan yoki massivlardan foydalaniladi. Jadval elementlari yagona nomga ega bo'lib, ular o'zlarining tartib nomeri (indeksi) bilan farq qiladi. Dasturlashda jadvalning nomi qaysi turga mansub bo'lsa jadvalning barcha elementlari ham shu turga tegishli bo'ladi.

Dasturlash tillarida jadval kattaliklardan foydalanish uchun jadvalni tavsiflash va kerakli jadval elementini ko'rsatish kerak bo'ladi.

Jadvalni tavsiflash jadvalning nomini, jadval elementlari sonini, turini, jadvalning o'lchovini aniqlashdan iboratdir.

Beysik tilida asosan bir yoki ikki o'lchovli jadval kattaliklardan foydalaniladi va jadval elementining quyi tartib nomeri har doim nol deb qaraladi. Bir o'lchovli jadvallarni tavsiflashda

$\langle N \rangle$ DIM $\langle \text{nom} \rangle (\langle \text{elementlar soni} \rangle)$

buyrug'idan, ikki o'lchovli jadvallarni tavsiflashda

$\langle N \rangle$ DIM $\langle \text{nom} \rangle (\langle \text{elementlar soni}, \text{elementlar soni} \rangle)$

buyrug'idan foydalaniladi. Bu yerda:

$\langle N \rangle$ -buyruq joylashgan satrning nomeri;

DIM- buyruqning nomi bo'lib, «o'lchov» degan ma'noni bildiradi;

$\langle \text{nom} \rangle$ - jadvalning nomi;

$\langle \text{elementlar soni} \rangle$ - indeksning yuqori chegarasi bo'lib, qiymati butun son, o'zgaruvchi yoki ifoda bo'lishi mumkin.

Misollar:

70 DIM X(10)

80 DIM A (5), B %(6), S \$(10)

90 DIM B (5,6), Q (N,M)

Dasturda DIM buyrug`ining uchrashi quyidagi ishlarni amalga oshiradi. Masalan, 70-satrdagi buyruq uchun jadvalning nomi X, jadval elementlarining turi haqiqiy, o`lchovi bir, elementlarining soni 11 ta bo`lib, elementlarining tartib nomeri 0 dan 10 gacha o`zgaradigan jadval uchun xotirasidan 11 ta joy ajaratadi:

X ₀	X ₁	X ₂	X ₃	X ₄	X ₅	X ₆	X ₇	X ₈	X ₉	X ₁₀	X-jadval
----------------	----------------	----------------	----------------	----------------	----------------	----------------	----------------	----------------	----------------	-----------------	----------

90-satrdagi buyruqda uchta jadval aniqlangan bo`lib, ularning turi mos ravishda haqiqiy, butun va harfiydir. 90-satrdagi buyruqda 2 ta ikki o`lchovli jadval aniqlangan bo`lib, nomlari V, Q elementlarining turi haqiqiy bo`lib, indeksleri B uchun 0 dan 5 gacha va 0 dan 6 gacha, Q uchun esa 0 dan N gacha va 0 dan M gacha o`zgaradi. V dagi elementlar soni 6\*7 ta, Q da esa (N+1)\*(M+1) ta, Q uchun N va M oldin aniqlangan bo`lishi kerak.

Jadval elementlarining tartib nomeri 10 dan oshmagan jadvalni Beysik tilida tavsiflamaslik ham mumkin.

Quyidagiga e`tibor bering:

70 DIM X(20), X%(20), X\$(20).

Bu yerda uchala jadvalning nomlari bir xilga o`xshab ko`rinadi, ular uchta jadval kattalikni tashkil qilib, turlari mos ravishda haqiqiy, butun va harfiydir.

Bu Beysik tilining yana bir muhim tomonlaridan biridir.

Jadval elementlariga mos ravishda X(I) yoki Y(I,J) ko`rinishda murojaat qilinadi va bunday o`zgaruvchilar indeksli o`zgaruvchilar deb ataladi. Mos ravishda X(10)- X nomli bir o`lchovli haqiqiy elementli jadvalning 10 nomerli elementini, X(I+1) esa I+1-elementini, Y(5,6)-esa Y nomli haqiqiy elementli jadvalning 5-nomerli satr va 6-nomerli ustunlarining kesishishida turgan elementini aniqlaydi. Indeksda ixtiyoriy ifoda turishi mumkin. Indeksning butun qiymati sifatida ifodaning butun qiymati olinadi. Bu qiymat indeksning yuqori tartib nomeridan oshmasligi kerak va quyi nomeridan kichkina bo`lmasligi kerak.

Dasturning ishlash jarayonida biror jadval kerak bo`lmay qolsa va bu jadval egallagan joyni boshqa bir maqsadda ishlatish zarur bo`lsa,

<N> ERASE <jadvalning nomi>

buyrug`idan foydalanish mumkin. Masalan,

100 ERASE A

nomli jadval egallagan joyni bo`shatadi. Bu buyruqdan keyin A nomli jadval kattalikning biror elementiga murojaat qilish xatolikka olib keladi.

Beysik tilida o`zgaruvchi va jadval elementlarini tozalash uchun

<N> CLEAR [<son>]

buyrug`idan foydalaniladi. Bu buyruqni dasturning boshida ishlatish maqsadga muvofiqdir. Masalan,

10 CLEAR

buyruq barcha kattaliklarning qiymatini tozalaydi.

Jadval kattaliklar ustida misollar bajarish

Elementlar soni n ta bo`lgan X jadvalning elementlarini kiritishda quyidagi dasturdan foydalanish qulaydir.

```
10 REM "JADVAL ELEMENTLARINI KIRITISH"  
20 INPUT "ELEMENTLAR SONI N=";N  
30 DIM X(N)  
40 FOR I=1 TO N  
50 PRINT "X( "; I; ")=";  
60 INPUT X(I)  
70 NEXT I
```

Jadval elementlarini chiqarishda ham yuqoridagi dasturdan foydalanish qulay. Buning uchun 60-satrni quyidagicha o`zgartirish yetarli:

```
60 PRINT X(I)
```

$X=(X_1, X_2, \dots, X_n)$ jadvalning musbat va manfiy elementlarini yig`indisi va sonini hisoblang.

```
10 REM YIG`INDI  
20 INPUT "ELEMENTLAR SONI N=";N  
30 DIM X(N)  
40 FOR I=1 TO N  
50 PRINT "X( "; I; ")=";  
60 INPUT X(I)  
70 NEXT I  
80 MUS=0: MAN=0: K_MUS=0: L_MAN=0  
90 FOR I=1 TO N  
100 IF X(I)>=0 THEN MUS=MUS+X(I): K_MUS=K_MUS+1 ELSE  
MAN=MAN+X(I): L_MAN=L_MAN+1  
110 NEXT I  
120 ? "MUSBAT YIG.="; MUS, "MUS. ELEM. SONI=";K_MUS  
130 ? "MANFIY YIG.="; MAN, "MAN. ELEM. SONI=";L_MAN  
140 END
```

$A(N,M)$ -ikki o`lchovli jadvalning eng katta va eng kichik elementini toping.

```

10 REM MAX_MIN
20 INPUT "N,M="; N, M
30 DIM A(N,M)
40 FOR I=1 TO N
50 FOR J=1 TO M
60 ? "A("; I; ", "; J; ")=": INPUT A(I,J)
70 NEXT J, I
80 MAX=A(1,1): MIN=A(1,1)
90 FOR I=1 TO N
100 FOR J=1 TO M
110 IF MAX<A(I,J) THEN MAX=A(I,J)
120 IF MIN>A(I,J) THEN MIN=A(I,J)
130 NEXT J, I
140 ? "MAX=",MAX, "MIN=";MIN
150 END

```

Nazariy savollar.

1. Jadval kattaliklar deganda nimani tushunasiz?
2. Beysik tilida asosan necha xil jadval kattaliklar bilan ishlash mumkin?
3. Beysik tilida massivlar qanday e`lon qilinadi?
4. Jadval kattaliklar ustida misollar bajarishga misollar ko`rsating.

Muammoli savollar

1. 40 DIM X(10)
50 DIM A(5), B%(6), S\$(101)
60 DIM B(5,6), Q(M,N)
- ushbu yozuvlarga izoh bering.
2. A(5,6) - matrisaning musbat elementlari yig`indisini topish algoritmi dasturini tuzing.
3. 10 DIM A(20)
20 For I=10 to 20
30 Input C(i): Next I
dastur xatosini aniqlang.
4. B(5) – massiv elementlarini kamayib borish tartibida joylashtiring.

3-Ma'ruza. Foydalanuvchi funksiyasi. Qism dastur. Foydalanuvchi funksiyalar va qism dasturlarini tashkil qilish. Funksiya va qism dasturlarga murojaat etish.

Reja:

1. Foydalanuvchining funksiyasi.
2. Qism dastur.

Foydalanuvchining funksiyasi.

Standart funksiyalarning ro'yxati har qanday funksiyani ham o'z ichiga olmaydi. Masalan, ctgx , arcsinx , arccosx , $\log_a x$ va h.k. Shuning uchun ham dastur tuzuvchi o'zining funksiyalarini aniqlashi yoki tavsiflashi mumkin. Bunday funksiyalar odatda **nostandard funksiyalar** yoki **foydalanuvchining funksiyalari** deb ataladi.

Foydalanuvchining funksiyasini dasturda aniqlash uchun quyidagi namunadan foydalaniladi:

$\langle N \rangle$ DEF FN $\langle \text{nom} \rangle$ ($\langle \text{argumentlar ro'yxati} \rangle$) = $\langle \text{funksiyaning ko'rinishi} \rangle$

Bunda,

$\langle N \rangle$ - funksiya aniqlangan satrning nomeri;

DEF (DEF ine) - aniqlash;

FN (FUNCTION) - funksiya degan ma'noni anglatadi;

$\langle \text{nom} \rangle$ - Beysik tilidagi o'zgaruvchining nomiga mos kelib, funksiyaning turini aniqlaydi;

$\langle \text{argumentlar ro'yxati} \rangle$ - bir-biridan vergul bilan ajratilgan o'zgaruvchilarning nomlari (ular kichik qavslar ichida bo'lishi shart) bo'lib, ularni **formal parametrlar** deb ataladi;

$\langle \text{funksiyaning ko'rinishi} \rangle$ - ifoda.

Misol,

1. $f(x) = x^2 + 1$ funksiyani Beysik tilida aniqlanishi:
10 DEF FNF (X) = X^2 + 1
2. $l(x, a) = \log_a x = \ln x / \ln a$
30 DEF FNL (X, A) = LOG(X) / LOG(A)
3. $s(x) = \arcsin x = \arctg(x / \sqrt{1 - x^2})$
20 DEF FNS (X) = ATN(X / SQR(1 - X^2))

Ma'lumki, lotin harflari 26 ta bo'lgani uchun bitta dasturda bir paytda 26 ta nostandard funksiyadan foydalanish mumkin.

Quyidagi ifoda orqali foydalanuvchining funksiyasini ishlatilishini ko'raylik:

$$S = \frac{\log_2 x + \log_5 (x^2 + 1)}{\lg x^2 + \log_7 (10 \cdot x)} \quad (3.1)$$

Agar biz ixtiyoriy a asosga ko`ra logarifmni

$$L(y, a) = \log_a y \quad (3.2)$$

ko`rinishida belgilasak, u holda (3.1) ni soddaroq qilib (3.2) yordamida quyidagicha yozish mumkin:

$$S = \frac{L(x, 2) + L((x^2 + 1), 5)}{L(x^2, 10) + L(10x, 7)} \quad (3.3)$$

Biz (3.2) ni foydalanuvchining funksiyasi ko`rinishida quyidagicha aniqlab olamiz:

$$\text{DEF FN L (Y,A)=LOG(Y) / LOG(A)}$$

Bu yerda LOG - Beysik tilida natural logarifmni bildiradi, Y va A formal parametrlar.

(3.1) ifodani foydalanuvchining funksiyasidan foydalanib hisoblash dasturi quyidagicha bo`ladi:

```
10 REM IFODA
20 INPUT "X=";X
30 DEF FNL (Y,A)=LOG(Y) / LOG(A)
40 S=(FNL(X,2)+FNL(X^2+1,5)) / (FNL(X^2,10)+FNL (10*X,7))
50 ? "S=";S
60 END
```

Nostandart funksiyalardan foydalanish dasturning hajmini kamaytirishga olib keladi. Bu esa dasturning ishlash tezligini ortishga olib keladi.

Qism dastur

Agar yuqorida ko`rsatilgan nostandart funksiyaga e`tibor bersak, unda faqatgina bitta ifoda turishi mumkin bo`lib, bu ifoda funksiyaning ko`rinishini aniqlaydi. Agar ifodalar soni bir qancha bo`lsa-chi yoki funksiyaning qiymati biror shartga bog`liq bo`lib, bir guruh buyruqlarning bajarilishiga bog`liq bo`lsa-chi?

Beysik tilida **qism dastur** tushunchasi kiritilgan bo`lib, bu yuqoridagi savollarga javob berish imkonini beradi.

Qism dastur deb dasturning shunday bir qismiga aytiladiki, unga dasturning hohlagan joyidan turib murojaat qilish mumkin.

Qism dasturga murojaat qilish uchun GOSUB buyrug`idan foydalaniladi. Uning ko`rinishi quyidagicha:

<N> GOSUB <M>

Bu yerda:

<N> - satrning nomeri;

GOSUB - buyruqning nomi, u GO (o`t) va SUB routine (qism dastur) so`zlaridan tashkil topgan bo`lib, qism dasturga o`tishni bildiradi;

<M> - qism dastur boshlangan satrning nomeri.

Misol:

30 GOSUB 100

Bu buyruq 100 - satrdan boshlangan qism dasturga murojaatni bildiradi.

Dasturda qism dasturlar bir qancha bo`lishi va bir qism dasturning ichidan turib boshqasiga ham murojaat qilish mumkin.

Qism dasturning oxirgi buyrug`i har doim

<K> RETURN

ko`rinishidagi buyruq bo`lishi shart. Bunda <K> - buyruq joylashgan satrning nomeri. RETURN - buyruqning nomi bo`lib, qaytish degan ma`noni bildiradi. Bu buyruq boshqarishni qism dasturga murojaat qilingan joyni o`ziga qaytaradi.

Misol:

$$S = \frac{4!+8!+n!}{(m!-p!)^2}$$

ni hisoblash dasturini tuzing.

Bu misolni qism dastur yordamida hisoblaymiz. Agar biz biror k faktorialni hisoblashning qism dasturini yoza olsak, u holda k ga (formal parametr) ketma-ket 4, 8, n , m va p (faktik parametrlar) qiymatlarni berib, kerakli faktoriallarni hosil qilish mumkin.

Endi k faktorialni hisoblash dasturini qism dastur shaklida yozamiz.

```
200 REM FAKTORIAL_QISM DASTUR
210 L=1
220 FOR I=1 TO K
230 L=L*I: NEXT I
240 RETURN
```

Bundan ko`rinib turibdiki, qism dasturning natijasi, ya`ni k faktorial L o`zgaruvchida hosil bo`lar ekan.

Endi asosiy dasturni yozamiz:

```
10 REM QISM DASTURDAN FOYDALANISH
20 INPUT "N,M,P="; N, M, P
30 K=4: GOSUB 200: S=L
40 K=8: GOSUB 200: S=S + L
50 K=N: GOSUB 200: S=S + L
60 K=M: GOSUB 200: T=L
70 K=P: GOSUB 200
80 S=S / (T-L)^2
90 ? "S=";S
100 END
200 REM FAKTORIAL_QISM DASTUR
210 L=1
220 FOR I=1 TO K
230 L=L*I: NEXT I
240 RETURN
```

Odatda qism dasturning matni END buyrug`idan keyin joylashgani ma`qul. Agar qism dastur matni END dan oldin joylashgan bo`lsa, u holda GOTO buyrug`i yordamida tashlab o`tish lozim, aks holda qism dastur dasturning tabiiy bajarilish ketma-ketligida bajarilib ketishi mumkin.

Nazariy savollar.

1. Foydalanuvchining funksiyasi deganda nimani tushunasiz?
2. Foydalanuvchining funksiyasiga misollar ko`rsating.
3. Qism dastur deganda nimani tushunasiz va ular nima uchun kerak?
4. Qism dasturga misollar ko`rsating.

Muammoli savollar

1. $f(x)=x^2+5x+6$ funksiya BEYSIK tilida qanday aniqlanadi.

2. $S(x) = \arccos x = (\pi/2) - \arctg(x/\sqrt{1-x^2})$ funksiya
20 DEF S(X)=PI/2-ATN(X/X^2)) ko`rinishida to`g`ri yozilganmi?
3. Formal parametrlarga ta`rif bering.
4. $S = \frac{4! \cdot 8! \cdot 19!}{(m! \cdot p!)^2}$ ni hisoblash dasturini tuzing.

m, p – berilgan sonlar.

4-Ma`ruza. Satr kattaliklarni qayta ishlash. Satr kattaliklarni qayta ishlovchi funksiyalar. Satr kattaliklarni qayta ishlovchi dasturlar tuzish.

Reja:

1. Satrli kattaliklarni qayta ishlovchi funksiyalar.
2. Matnning uzunligini aniqlash.
3. Chapdan, o`ngdan va matnning orasidan ajratish.
4. Matnning ichidan boshqa matnni qidirish.
5. Maxsus funksiyalar.

Kompyuterlarning muhim imkoniyatlaridan yana biri uning matnlarni qayta ishlay olishidir. Matnli ma`lumotlarni qayta ishlash deganda ularni bir-biriga ulash, uning ma`lum qismini ajratib olish, ulardagi belgilarni, boshqa belgilar bilan almashtirish va shunga o`xshash masalalarni tushunamiz.

Satrli kattaliklarni qayta ishlovchi funksiyalar.

1. ***Matnning uzunligini aniqlash.*** Matnning uzunligi deganda undagi belgilar soni tushuniladi va u

LEN(X\$)

yordamida hisoblanadi. Bu yerda X\$- matnli kattalik. Masalan,

LEN("KITOB")=5

Agar, A\$="A+B" bo`lsa, LEN(A\$)=3 bo`ladi. Bo`sh matnning uzunligi nolga teng bo`ladi. U "" ko`rinishida bo`ladi.

2. ***Chapdan ajratish.*** Ixtiyoriy matnning chapdan biror bo`lagini ajratib olish uchun

LEFT\$(X\$, N)

funksiyasidan foydalaniladi. Bunda X\$ ixtiyoriy matnli kattalik, N esa nechta belgi ajratilishi kerakligini ko`rsatuvchi son.

Masalan,

$$\text{LEFT\$}(\text{"ALIFBO"}, 4) = \text{"ALIF"}$$

Bu funksiyaning natijasi yana matn bo`ladi.

$$X\$ = \text{"Mehribon"}, N = 5$$

uchun

$$\text{LEFT\$}(X\$, N) = \text{"Mehri"}$$

bo`ladi.

3. **O`ngdan ajratish.** Ixtiyoriy matnning oxiridan (o`ngdan) biror bo`lagini ajratib olish uchun

$$\text{RIGHT\$}(X\$, N)$$

funksiyasidan foydalaniladi. Bu yerda ham X\$ va N parametrlar LEFT\$ dagidek. Masalan,

$$\text{RIGHT\$}(\text{"ALIFBO"}, 4) = \text{"IFBO"}$$

Bu funksiyaning natijasi ham yana matn bo`ladi.

$$X\$ = \text{"Mehribon"}, N = 5$$

uchun

$$\text{RIGHT\$}(X\$, N) = \text{"ribon"}$$

bo`ladi.

4. **Orasidan ajratish.** Ixtiyoriy matnning orasidan biror bo`lagini ajratib olish uchun

$$\text{MID\$}(X\$, K, L)$$

funksiyasidan foydalaniladi. Bu yerda

X\$ - matn;

K - qaysi o`rindaligini ko`rsatuvchi son;

L - esa K dan boshlab nechta belgi ajratish kerakligini ko`rsatuvchi son.

Masalan,

$$\text{MID\$}(\text{"MEHRIBON"}, 4, 3) = \text{"RIB"}$$
$$\text{LEFT\$}(\text{"ALIFBO"}, 4) = \text{"ALIF"}$$

Bu funksiyaning natijasi matn bo`ladi.

$$X\$ = \text{"Mehribon"}, K = 3, L = 2$$

uchun

MID\$(X\$,K,L)="hr"

bo`ladi.

Yuqoridagi funksiyalardan foydalanib, matnlar orasiga yangi matnlar kiritish mumkin.

MID\$ funksiyasining yana bir xususiyati shundaki, uning yordamida matnning biror qismiga to`g`ridan- to`g`ri yangi matn jo`natish mumkin.

MID\$(X\$,K,L)=A\$

bu matnning K-o`rnidan boshlab L ta belgisi o`rniga A\$ matni joylaydi. A\$ matnning uzunligi ajratilgan joyning L uzunligidan oshmasligi kerak.

Misol:

Uzunligi 3 dan kichik bo`lmagan kiritilgan ixtiyoriy matnning uzunligini ko`rsatuvchi, uning chap tomonidan 3 ta, o`ng tomonidan 3 ta, orasidan ikkinchi belgidan boshlab 2 ta ajratuvchi dastur tuzing.

```
10 REM "SATRLI FUNKSIYALAR"
20 INPUT "MATNNI KIRITING S$=";S$
30 L=LEN(S$); ? "MATNNING UZUNLIGI =" ;=L
40 IF L<3 THEN GOTO 80
50 L$=LEFT$(S$,3): ? "MATNNING CHAPDAN 3 TASI =" ;L$
60 R$=RIGHT$(S$,3): ? "MATNNING O`NGDAN 3 TASI =" ;R$
70 M$=MID$(S$,2,2): ? "MATNNING 2 BEL. BOSH.2TA =" ;M$
80 END
```

5. *Matnning ichidan boshqa matnni qidirish.* Buning uchun

INSTR (N, <berilgan matn>, <qidirilayotgan matn>)

funksiyasidan foydalaniladi. Bu yerda N-qidirishni berilgan matnning nechanchi belgisidan boshlash kerakligini ko`rsatuvchi son.

Agar N=1 bo`lsa, u holda bu funksiyani

INSTR (<berilgan matn>, <qidirilayotgan matn>)

ko`rinishida ham ishlatish mumkin. Bu funksiyaning natijasi butun son bo`lib, u qidirilayotgan matn berilgan matnda nechanchi o`rindan boshlanganligini ko`rsatadi. Agar qidirilayotgan matn berilgan matnda uchramasa, funksiyaning qiymati nolga teng bo`ladi.

Misol:

```
INSTR(4,"PAXTAKOR", "KO")=6
INSTR("PAXTAKOR", "FF")=0
```

INSTR(2,"INFORMATIKA", "I")=9

Berilgan ixtiyoriy matndagi birinchi uchragan "BA" ning o`rnini toping.

```
10 REM BA
20 INPUT"MATNNI KIRITING=";X$
30 K=INSTR(X$,"BA")
40 IF K=0 THEN 60
50 ? "VA"; K; "DAN JOYLASHGAN": END
60 ? "BA"; "MATNDA YO`Q"
```

6) *Probellarni (bo`sh joylarni) hosil qilish.*

Faqat probellardan tashkil topgan matnni hosil qilishda
SPACE\$(N)

funksiyasidan foydalaniladi. Bu yerda N-probellar soni. Masalan,
B\$=SPACE\$(10)

buyruq B\$ ga 10 ta probeldan iborat bo`lgan matnni joylashtiradi. Biror matnni ekranning o`rtarog`ida hosil qilishda bu funksiya juda qo`l keladi:

```
30 ? SPACE$(20); "SALOM"
```

Faqat PRINT buyrug`ida qo`llanilishi mumkin bo`lgan va probellarni hosil qiluvchi

SPC(N)

funksiya mavjud. Bu funksiyaning TAB(N) funksiyasidan farqi shundaki, TAB(N) da N satrdagi o`rnini ko`rsatsa, SPC da N probellar sonini ko`rsatadi. Bu funksiyalarning farqini quyidagi dastur yordamida ko`rish mumkin:

```
10 FOR I=1 TO 5
20 ? TAB(I); "A";
30 NEXT
40 FOR I=1 TO 5
50 ? SPC(I); "A";
60 NEXT
```

Bu dasturning natijasi quyidagi ko`rinishda bo`ladi:

```
AAAAA
_A_A_A_A_A
```

7) **Bir xil belgilar ketma-ketligini hosil qilish.** Bunda

STRIG\$(N,"<belgi>")

funksiyasidan foydalaniladi. Masalan,

STRIG\$(10, "*")

Funksiyasi 10 ta yulduzcha hosil qiladi. "<belgi>" o`rnida uning kodini ham yozish mumkin. Masalan,

STRING\$(5,"A") va STRING\$(5,65)

teng kuchlidir. Funksiyadagi <belgi> o`rniga belgilar ketma-ketligi olinsa, u holda funksiya faqat birinchisidan tashkil topgan belgilar ketma-ketligini hosil qiladi. Misol: Uzunligi 20 dan oshmagan 5 ta so`zni o`ng tomondan tekislab chiqing.

```
10 REM "O`NGDAN TEKISLASH": DIM X$(5)
20 FOR I=1 TO 5
30 INPUT "MATNNI KIRITING="; X$(I): NEXT
40 CLS
50 FOR I=1 TO 5
60 S=35-LEN(X$(I))
70 ? I; STRING$(S,"-"); X$(I)
80 NEXT
90 END
```

8) **Maxsus funksiyalar.** Matndan songa o`tkazishda

VAL ("MATN")

funksiyasidan foydalaniladi. Bu funksiya "MATN"dan songa o`tish mumkin bo`lganda mos songa o`tkazadi, aks holda bu funksiyaning qiymati nolga teng bo`ladi. Masalan,

```
VAL("4")=4
VAL("1.3")=1.3
VAL("-2A")=-2
VAL("A")=0
VAL("$B1111111")=255
VAL("$O377")=255
VAL("$HFF")=255
```

Sondan matnga o`tkazishda STR\$(<ifoda>) funksiyasidan foydalaniladi. Bu funksiya arifmetik ifodaga mos matnni hosil qiladi.

Masalan,

STR\$(25)="25"

STR\$(-2.5)="-2.5"

P=12 uchun

STR\$(P)="12"

Belgiga mos kodni topishda

ASC("<belgi>")

funksiyasidan foydalaniladi. Bu funksiya belgiga mos keladigan kodni hosil qiladi. Masalan,

ASC("A")=65

Bu funksiyada "<belgi>" o`rnida satrli o`zgaruvchi ham kelishi mumkin.

P\$="P"

ASC(P\$)

Kodga mos belgini aniqlashda

CHR\$(<ifoda>)

funksiyasidan foydalaniladi. Bu funksiya <ifoda>ga mos keladigan belgini hosil qiladi. Masalan,

CHR\$(65)="A"

Kodlarga mos keladigan belgilarni aniqlashda quyidagi dastur juda qo`l keladi.

10 FOR K=1 TO 255

20 ? K; TAB(10); CHR\$(K)

30 NEXT

O`nlik sanoq sistemasidan ikkilik, sakkizlik va o`n oltilik sanoq sistemalariga o`tkazishda mos ravishda

BIN\$(N), OCT\$(N), HEX\$(N)

funksiyalaridan foydalaniladi. Masalan,

? "20 ";BIN\$(20), OCT\$(20), HEX\$(20)

buyruq bajarilganda ekranda:

20 10100 24 14

hosil bo`ladi.

Misol:

```
10 REM "TARJIMA"  
20 FOR I=0 TO 100  
30 ? "10=";I, "2=";BIN$(I), "8="; OCT$(I), "16="; HEX$(I)  
40 NEXT
```

Ikkilangan aniqlikdagi sonlarni oddiy aniqlikka o`tkazish uchun CSNG (X)funksiyasidan foydalaniladi.

Masalan,

CSNG(1234.1234)=1234.12

CSNG(123456789)=123457000

Nazariy savollar.

1. Satr kattaliklarni qayta ishlovchi funksiyalarni sanab bering.
2. «KOMPYUTER» so`zidagi belgilar sonini aniqlang. Shu so`zning chapdan, o`ngdan va orasidan 3 tadan belgilarni kesib olish dasturini tuzing.
3. Probellarni hosil qilishga misollar ko`rsating.
4. Maxsus funksiyalar ishtirokida dasturlar tuzing.

Muammoli savollar

1. A\$= «Inson». B\$= «Format», S\$= «Matematika» belgili ma`lumotlardan Informatika so`zini tuzish dasturini tuzing.
2. Bo`sh joylarni (probellarni) hosil qilish qanday amalga oshiriladi.
3. VAL("4")=; VAL("-2A")=; VAL("\$HFF")=; STR\$(25)=; STR\$(P)= natijalarini yozing.
4. ? "40"; BIN\$(40), OCT\$(40), NEX\$(40) – buyruq natijalari qanday bo`ladi.

5-Ma'ruza Beysik (Paskal) tilining imkoniyatlari. Dasturni xotiraga yozish va undan o`qish. Beysik (Paskal) tilining buyruqlari va ularning qo`llanishi.

Reja:

1. Dasturni tashqi xotiraga yozish va undan o`qish.
2. Tanlash buyruqlari.
3. Kiritish buyruqlari.
4. Qolipga moslab chiqarish.
5. Belgining ekrandagi o`rnini aniqlash.

Ayrim hollarda ishlagan dasturdan keyinchalik ham foydalanish ehtiyoji tug`iladi. Bunday hollarda dasturning matnini tashqi hotirada (magnit disklarda) saqlab qolish maqsadga muvofiqdir. Buning uchun

SAVE "HOM"

buyrug`idan foydalanish kerak. Bu yerda «NOM» faylning nomi. Bu buyruq nomersiz bo`lib, shu ondayoq bajariladi va magnit diskga xotiradagi Beysik dasturning matnini «NOM» bilan yozib qo`yadi. Dasturning yozilganligiga ishonch hosil qilish uchun

FILES

buyrug`i yordamida tekshirish mumkin. Bu buyruq magnit diskdagi dasturlarning ro`yxatini chiqaradi.

Dasturni tashqi xotiradan ichki xotiraga chaqirish uchun

LOAD "HOM"

buyrug`idan foydalaniladi. Dasturni ichki xotiradan tashqi xotiraga yozishda undagi dastur saqlanadi. Agar tashqi xotirada «NOM» ismli dastur bo`lib, biz yana shu «NOM» ismi bilan yangi dasturni saqlasak, u holda eski dastur yo`qolib, uning o`rniga shu nom bilan yangi dastur hosil bo`ladi. Tashqi xotiradan chaqirilgan dasturni ishlatish uchun

RUN

buyrug`idan foydalaniladi. Dasturni tashqi xotiradan chaqirib birdaniga ishlatish uchun

LOAD "HOM", R

buyrug`idan foydalanish mumkin.

Xotirada turgan dastur matnini qog`ozga chop qilish uchun

LLIST

buyrug`idan foydalanish mumkin. Bu buyruq ham oldingi ma`ruzalarimizda ko`rib o`tganimiz LIST buyrug`i kabi ishlaydi.

Tanlash buyruqlari

Biz oldingi ma`ruzalarimizda shartsiz o`tish (GOTO) va qism dasturga o`tish (GOSUB) buyrug`lari bilan tanishib o`tgan edik. Bu buyruqlar har doim aniq bir joyga o`tishni ta`minlaydi. Masalan,

30 GOTO 100

yoki

30 GOSUB 100

buyruqlari bajarilganda 100-nomerli satrga o`tishni ta`minlaydi. Ammo bu buyruqlarni quyidagi ko`rinishda ham ishlatish mumkin:

<N> ON <ifoda> GOTO <N1>, <N2>,. . .,<NK>

<N> ON <ifoda> GOSUB <N1>, <N2>,. . .,<NK>

Bunda

<N>-buyruq joylashgan satrning nomeri;

<N1>, <N2>,. . .,<NK>-lar o`tilishi kerak bo`lgan satrlarning nomerlari.

Bu buyruqlar bajarilganda <ifoda> ning qiymati tekshiriladi. Agarda u birga teng bo`lsa, mos ravishda <N1> satrga, ikkiga teng bo`lsa, <N2> satrga va h.k. Qolgan hollarda, ya`ni bu ifodaning qiymati birdan kichik yoki K dan katta bo`lganda, bu buyruqlar joylashgan satrdan keyingi satrga o`tiladi.

Misol:

```
10 REM MISOL
20 INPUT "K=";K
30 ON K GOSUB 50, 60, 70
40 END
50 ? K
60 ? K^2
70 ? K^3
```

Bu dasturning tekshirilishini sizning o`zingizga havola qilamiz.

Kiritish buyruqlari

INPUT buyrug`idan tashqari Beysik tilida boshqa kiritish buyruqlari ham qo`llaniladi. Ular bilan tanishib o`tamiz.

<N> LINE INPUT ["<izoh>";] <nom>

Bu buyruqning INPUT buyrug`idan farqi shundaki, buyruq bajarilganda ekranda so`roq belgisi hosil bo`lmaydi va kiritilayotgan qiymatlar matn deb tushuniladi. Shuning uchun ham <nom> faqat bitta matnli o`zgaruvchining nomi bo`lishi mumkin.

O`zgaruvchining qiymati sifatida <Enter> tugmasini bosilgunga qadar kiritilgan belgilar olinadi.

Bu buyruq yordamida sonli qiymat kiritishga to`g`ri kelsa VAL funksiyasidan foydalanish kerak. Masalan,

```
10 LINE INPUT "X NING QIYMATI=";X$
20 X=VAL(X$)
30 ? X^2, X$
40 END
```

Berilganlarni kiritishning yana bir yo`li

INPUT\$(K)

buyrug`idir. Bu yerda K-kiritilayotgan belgilar soni. Bu buyruqda ham kiritilayotgan belgilar matnli o`zgaruvchilarning qiymati deb qaraladi. Tugmalar orqali terilgan belgilar ekranda hosil bo`lmaydi va <Enter> tugmachasini bosishga ehtiyoj yo`q. Masalan,

K\$=INPUT\$(3)

buyruq o`zgaruvchiga tugmachalar orqali terilgan 3 ta belgini taqdim etadi.

Tugmachalar orqali kiritishning yana bir yo`li

INKEY\$

buyrug`i bo`lib, bu buyruq biror tugmachani bosilgan yoki bosilmaganligini aniqlashda muhim rol o`ynaydi. Tugmachalar bosilishi bilan bu buyruq belgiga mos kodni hosil qiladi, aks holda bu buyruqning qiymati nol kod yoki bo`sh satr bo`ladi. Masalan,

```
10 REM BELGINING KODINI HOSIL QILISH
20 K$=INKEY$: IF K$="" THEN 20
30 ? K$; "NI KODI ="; ASC(K$)
40 ? "DAVOM ETASIZMI Y/N"
50 K1$=INKEY$
60 IF K1$="Y" THEN GOTO 20
70 IF K1$="N" THEN 80 ELSE GOTO 50
80 END
```

Qolipga moslab chiqarish

Beysik tiliga yana bir chiqarish buyrug`i kiritilgan bo`lib, u qiymatlarni ma`lum qolipga (formatga) moslab, chiqarish imkonini beradi. Bu buyruqning ko`rinishi

<N> PRINT USING "<qolip>"; <ifodalar ro'yxati>

Bu yerda

N –buyruq joylashgan satrning nomeri;

PRINT USING-buyruqning nomi bo'lib, qolipga moslab chiqarishni bildiradi;

"<qolip>"-qiymatning ko'rinishini aniqlovchi satr bo'lib, u qo'shtirnoqlar ichida turishi shart;

<ifodalar ro'yxati> - PRINT buyrug'idagidek.

1. Sonli kattaliklar uchun qolip tanlash. Sonli kattaliklarning qiymatlarini chop qilishda quyidagi qolip belgilaridan foydalaniladi:

- o'nli raqamlarni aniqlash belgisi;

. – sonning butun va kasr qismini aniqlash belgisi;

^ - sonni o'nning darajalari (E) ko'rinishida chiqarish belgisi;

+ - ishorani aniqlash belgisi;

- - ishorani aniqlash belgisi;

** - to'ldiruvchi belgisi;

\$\$ - pul (dollar) belgisi.

Qolipdagi har bir # belgisi chop qilinayotgan sondagi bitta raqamga mos keladi. Masalan,

10 PRINT USING "##"; 9,25

buyruq natijasi

_925

ko'rinishida bo'ladi. Demak har bir chiqarilayotgan son uchun belgidagi qolip ortiqcha bo'lsa, u chapdan probel () bilan to'ldirilgan ekan, ajratilgan joy yetarli bo'lmasa, u holda sonning oldiga % belgisi qo'yib, qolipsiz (PRINT buyrug'idagidek) chiqariladi. Masalan,

20 PRINT USING "###"; 5,-5,144,-144,1235

buyruqning natijasi quyidagicha bo'ladi:

__5 _-5 144 %-144 %1235

bundan ko'rinib turibdiki, har bir son uchun qolipda uning ishorasi uchun ham joy ajratish lozim ekan. Kasr sonlarni chiqarishda sonning butun qismini kasr qismidan ajratuvchi o'nlik nuqtaning o'rmini, butun va kasr qismlaridan nechtadan raqam olish kerakligini qolipda ko'rsatish kerak bo'ladi. Masalan,

30 PRINT USING "##.##"; 3.1415, 25.236, 14.2 –5,
-34.12, 142

buyruqning natijasi quyidagicha bo`ladi:

3.1425. 2414.20 -5.00 %-34.12 %142.00

Sonning ishorasini chiqarish uchun har doim «+» belgisini qolipda birinchi o`rinda ko`rsatish kerak. Masalan,

40 PRINT USING "+###";344, -34

natijasi

+ 344 -34

ko`rinishda bo`ladi.

Ayrim hollarda juda katta yoki juda kichik sonlarni eksponensial ko`rinishda chiqarish maqsadga muvofiq bo`ladi. Masalan, 1000 000 000 sonini 0.10E+10 ko`rinishida uchun qolipni quyidagicha tanlash lozim bo`ladi:

40 PRINT USING "#.## ^ ^ ^ ^"; 1000000000

Qoliplar orasiga qo`yilgan probellar soni chiqarilayotgan qiymatlar orasidagi probellarni aniqlaydi.

2. **Matnli kattaliklar uchun qoliplar tanlash.** Sonlardagidek matnlarni ham ma`lum bir qolipga solib chiqarish mumkin. Bu qolipda asosan quyidagi belgilar ishlatilishi mumkin:

! – matnning birinchi belgisini chiqarishda qo`llaniladi.

Masalan,

10 ? USING "!"; "SALOM"

buyrug`i ekranda S ni hosil qiladi. Xuddi shuningdek

10 ? USING "!"; "SALOM";"ELKAN";"NON"

buyrug`i ekranda

SEN

matnini hosil qiladi. Sonli qoliplardagidek bu yerda ham qolip ro`yxatdagi barcha elementlarga tegishli bo`ladi.

&- yangi matnni ulash imkonini beradi. Masalan,

10 ? USING "&"; "SALOM"; "KATTA"

buyruqning natijasi

KATTA SALOM

ko`rinishida bo`ladi. Agar ro`yxatda bir qancha element bo`lsa, u holda qolipdagi matn har bir so`zga ulanadi. Masalan,

10 ? USING "&KASH"; "VAFO"; "JAFO"

buyruqning natijasi

VAFOKASH JAFOKASH

ko`rinishida bo`ladi.

3. \<n ta probel> - bu qolip matndan birinchi $n+2$ ta belgini ajratib chiqarishni ta'minlaydi. Agar $n=0$ bo`lsa, \ ko`rinishida bo`lsa, matnning birinchi ikkita belgisi chiqariladi.

Quyidagi dasturga e'tibor bering:

```
10 REM QOLIP
20 FOR I=1 TO 3: READ A$
30 ? USING "!"; A$
40 ? USING "\ "; A$
50 ? USING "\__ "; A$
60 ? USING "\&_YUTDI "; A$
70 ? STRING (20, "-")
80 NEXT I
90 DATA "PAXTAKOR", "BINOKOR", "OCHKO`Z"
```

Bu buyruqning natijasi quyidagicha bo`ladi:

P
PA
PAXT
PAXTAKOR YUTDI

B
BI
BINO
BINOKOR

O
OCH
OCHKO`
OCHKO`Z YUTDI

Belgining ekrandagi o`rnini aniqlash


Kompyuter ekranidagi maxsus belgining hosil bo`ladigan o`rnini ko`rsatuchi maxsus belgi **yurgich** (*kursor*) deb ataladi. Yurgichning ko`rinishi turli kompyuterlarda

turlicha bo`ladi. Kompyuter ekrani matnli holatda bo`lganda ekranda tugmachalardagi ixtiyoriy belgini hosil qilish mumkin. Ekranning o`lchami oldindan aniqlangan bo`ladi.

Ektranda kursorning ixtiyoriy o`ringa kelishi uchun Beysik tilida

<N> LOCATE X,Y, [K]

buyrug`idan foydalaniladi. Bu yerda X- kursorning X o`q bo`yicha tutgan o`rni , Y- kursorning Y o`q bo`yicha tutgan o`rni, K - kursorni ko`rinish ko`rinmasligini ta`minlovchi son (K=0 da ko`rinmaydi, K=1 da ko`rinadi).



Bu yerda $0 \leq X \leq 80$, $0 \leq Y \leq 24$

Kursorning ekrandagi o`rnini aniqlash uchun SSRLIN (Y bo`yicha), POS (X bo`yicha) funksiyalaridan foydalaniladi.

Quyidagi dastur ekranning har xil joylarida SALOM degan so`zni hosil qiladi.

```
10 REM LOCATE
20 WIDTH 80
30 FOR I=1 TO 30
40 X=INT(RND(1)*39): Y=INT(RND(1)*24)
50 LOCATE X,Y,1: "CALOM"
60 NEXT I
```

Nazariy savollar.

1. Dasturni qanday qilib tashqi xotiraga yoziladi?
2. Tanlash buyruqlarini sanab o`ting.
3. Kiritish buyruqlaridan qanday xollarda qo`llaniladi?
4. Qolipga moslab chiqarish buyruqlari ishtirokida misollar ko`rsating.

Muammoli savollar

1. Matnni tashqi xotirada saqlash qanday amalga oshiriladi?
2. FILES, LOAD, LLIST – vazifalarini tushuntiring.
3. 20 PRINT USING "###"; 5, -5.144, -144.1235 – buyruq natijalarini ko`rsating.
4. Ekranni turli xil joylarda «ILM» so`zini hosil qilish dasturini tuzing.

6-Ma'ruza. Beysik (Paskal) tilining grafik imkoniyatlari. Funksiya grafigini belgilar yordamida tasvirlash. Funksiyani xisoblash va grafigini ekranda tasvirlash dasturlarini tuzish.

Reja:

1. Beysik tilining grafik imkoniyatlari.
2. Funksiya grafigini belgilar yordamida tasvirlash.
3. Grafik xolatga o'tish.

Beysik tilining yana bir imkoniyatlaridan biri grafik imkoniyatidir, ya'ni natijalarni grafik yoki chizma shaklida tasvirlashdir. Ma'lumki, natijalarni har doim sonli yoki matnli ko'rinishda tasvirlash maqsadga muvofiq emas. Chunki bu sonlardan ma'lum bir xulosa chiqarish uchun ularni yana bir qayta ishlash kerak bo'ladi. Agar shu ma'lumotlar biror chizma shaklida tasvirlansa, foydalanuvchilar uchun bu juda qulay bo'lib, xulosa chiqarish uchun oson bo'ladi.

Shuning uchun ham axborotlarni chizma yoki gistogrammalar shaklida tasvirlash muhim ahamiyat kasb etadi. Beysik tili bu imkoniyatga egadir.

Grafikni belgilar orqali tasvirlash

Chizmani uzluksiz chiziq shaklida tasvirlash ancha qiyin masala bo'lib, u maxsus grafik holatlarga ega bo'lgan kompyuterlarda amalga oshiriladi. Lekin bu murakkablikni grafikning ayrim olingan belgilarining o'zni sifatida qarab, osongina hal qilish mumkin va grafikning ko'rinishi yoki o'zgarishi haqida xulosa chiqarish imkonini beradi.

Faraz qilaylik $y = f(x)$ funksiyaning grafigini ayrim olingan belgilarning o'zni sifatida tasvirlash talab qilinsin. Belgilar sifatida nuqta yoki yulduzchalar olish ham mumkin.

Ma'lumki, funksiyaning grafigini chizishda, oldin funksiya aniqlangan va grafik chizishda kerak bo'ladigan $[a,b]$ oraliq aniqlab olinadi va uni shu oraliqdagi qiymatlar jadvali tuzib olinar edi.

Agar $f(x) = x^2$ deb olsak, bu funksiya uchun qiymatlar jadvali tuzamiz. Berilgan oraliq $[-5,5]$ bo'lsin. Oraliqni 10 bo'lakka bo'lib, qadamni $h=1$ deb olib, funksiyaning qiymatlar jadvalini hosil qilamiz.

x_i	-5	-4	-3	-2	-1	0	1	2	3	4	5
y_i	25	16	9	4	1	0	1	4	9	16	25

Hosil bo'lgan bu qiymatlarni koordinata sistemasiga qo'yib chiqamiz.

Bizga ma'lumki, kompyuter ekranining o'rinlari soni juda chegaralangan va ular faqatgina butun sonlarnigina qabul qila oladi.

Berilgan funksiyaning grafigini "*" orqali tasvirlash dasturini ko'rib chiqamiz. Funksiyaning $[-5,5]$ oraliqdagi qiymatlari 25 dan oshmagani uchun masshtab tanlaymiz.

```


10 REM F(X)
20 X=-5
30 ? “^ y”
40 ? “!”: ?”!”
50 IF X=0 THEN 90
60 ? “!”; TAB(X^2=1);”*”
70 X=X+1: IF X<=5 THEN 50
80 END
90 ? “*”;STRING$ (30,”-”); “> x”
100 GOTO 70

```

Yuqoridagi dastur bajarilishi natijasida kompyuter ekranida 6.1-rasmdagi kabi tasvir hosil bo`ladi.

Funksiyaning grafigini bunday usulda tasvirlash unchalik aniq natija bermaydi, lekin ba`zi bir funksiyalarning qiymatlari asosida xulosa yasash uchun yetarlidir.

Grafiklarni kompyuterning grafik imkoniyatlaridan foydalanib tasvirlash uchun kompyuterning grafik ekranidan foydalaniladi. Kompyuterning grafik va matnli ekranidan bir paytda foydalanish mumkin emas.



6.1-рasm

Qaysi ekrandan foydalanish dasturchining tanlashiga bog`liq. Har bir ekran o`zining o`lchamiga ega bo`lib, u ekrandagi nuqtalar soniga qarab aniqlanadi va har bir ekrandagi nuqtalar ham o`zining o`lchamiga ega. Bu o`lcham *piksel* deb ataladi.

Ekranlarni tanlash

<N> SCREEN K

buyrug`i yordamida aniqlanadi. Bu buyruqda K quyidagi qiymatlarni qabul qilishi mumkin:

K=0 - 40×24 o`lchamli matnli ekran

K=1 - 32×24 o`lchamli matnli ekran
 K=2 - 256×192 o`lchamli yuqori holatli grafik ekran
 K=3 - 64×48 o`lchamli quyi holatli grafik ekran
 K=4 - 256×192 o`lchamli yuqori holatli grafik ekran
 K=5 - 256×212 o`lchamli yuqori holatli grafik ekran
 K=6 - 512×212 o`lchamli yuqori holatli grafik ekran
 K=7 - 512×212 o`lchamli yuqori holatli grafik ekran
 K=8 - 256×212 o`lchamli yuqori holatli grafik ekran

Rang tanlash. Biror bir nuqtani yoki shaklni ekranda hosil qilish uchun uning rangi tanlanadi. Shaklning rangini tanlashda

$\langle N \rangle \text{ COLOR } \langle P \rangle, \langle L \rangle, \langle K \rangle$

buyrug`idan foydalaniladi. Bunda $\langle N \rangle$ - buyruq joylashgan satrning nomeri; $\langle P \rangle$ - tasvirning rangi; $\langle L \rangle$ - ekranning rangi; $\langle K \rangle$ - ekran chegarasining rangi (hoshiya). $\langle P \rangle, \langle L \rangle, \langle K \rangle$ lar 0 dan 15 gacha bo`lgan qiymatlarni qabul qilishi mumkin.


Ekranda biror rangdagi nuqtani hosil qilish uchun

$\langle N \rangle \text{ PSET}(x,y), \langle P \rangle$


buyrug`idan foydalaniladi. Bu buyruq koordinatasi (x,y) bo`lgan nuqtani $\langle P \rangle$ rangda ekranda hosil qiladi.

Misol,

PSET(20,20),6



Ko`p hollarda biz tabiiy koordinata sistemasida ishlashga odatlanganmiz. Kompyuterdagi koordinata sistemasidan tabiiy koordinata sistemasiga o`tish uchun $u=191-u$ bilan belgilanadi.



PSET buyrug`i kabi ishlovchi yana bir buyruq PRESET buyrug`idir.

<N> PRESET (x,y), <P>

Bu buyruqning PSET buyrug`idan farqi shundaki, agar unda rang ko`rsatilmasa, ekranda hosil bo`layotgan nuqta ekran rangida hosil bo`ladi, go`yoki nuqta ekrandan yo`qotiladi.

Quyidagi dastur ekranda nuqtalarni turli rangda hosil qiladi. Rang va nuqtalar RND funksiyasi yordamida hosil qilinadi.

```
10 REM KALEYDOSKOP
20 SCREEN 2
30 FOR I=1 TO 50
40 X=INT(RND(1)*256)+1:Y=INT(RND(1)*191)+1
50 P=INT(RND(1)*15)+1
60 PSET(X,Y),P
70 NEXT I
80 K$=INKEY$: IF K$=" " THEN 80
```

Nazariy savollar.

1. Ekranni grafik holatga o`tkazuvchi buyruqning to`la formatini ko`rsating.
2. Rang tanlash buyrug`i ishtirokida misollar ko`rsating.
3. Ekranda har xil rangdagi nuqtalarni hosil qiluvchi dastur tuzing.
4. $y=x^3$ funksiyaning grafigini belgilar orqali tasvirini chiqarib bering.

Muammoli savollar

1. Ekran turlarini sanab bering.
2. $y=x^4$ funksiya grafigini belgilar orqali tasvirga chiqarib bering.
3. <N> COLOR <P>, <L>, <K> buyruqni izohlab bering.
4. Ekranda har xil rangdagi nuqtalarni tasvirlash dasturini tuzing.

7-Ma'ruza. Ekranda turli grafik shakllarni hosil qilish. Kema, to'g'ri to'rtburchaklar chizish. Ekranda turli grafik shakllarni hosil qilish uchun dasturlar tuzish. Aylana, ellips va ularni yoylari chizish.

Reja:


1. Kesmalar chizish.
2. To'g'ri to'rtburchaklar chizish.
3. Ichi bo'yalgan to'g'ri to'rtburchaklar chizish.
4. Aylana, ellips va ularning yoylarini chizish.
5. Ellips sektori va ellips yoyi chizish.
6. Aylana, ellips va shar figuralari uchun dasturlar tuzish.

Kema va to'g'ri to'rtburchakni chizish uchun maxsus buyruq mavjud bo'lib, u quyidagi ko'rinishga ega:

a) **Kesmani chizish.**

$\langle N \rangle$ LINE $(X_1, Y_1)-(X_2, Y_2)$, $\langle P \rangle$

Bu yerda (X_1, Y_1) va (X_2, Y_2) kema uchlarining koordinatalari.



Bu buyruq bajarilganda ekranda uchlari (X_1, Y_1) va (X_2, Y_2) nuqtalarda bo'lgan kema $\langle R \rangle$ rangda hosil bo'ladi. Rangni ko'rsatmaslik ham mumkin. Misol,

```
10 CLS:SCREEN 2
20 REM KESMA
30 LINE (30,20)-(120,70),6
40 REM GORIZONTAL KESMA
50 LINE (10,10)-(100,10)
60 REM VERTIKAL KESMA
70 LINE (20,20)-(20,130)
80 GOTO 80
```

Bu buyruq yordamida koordinata o'qlarini ham tasvirlash mumkin.

```
10 CLS:SCREEN 2
20 LINE (128,0)-(128,191)
30 LINE (0,96)-(255,96)
40 GOTO 40
```

LINE buyrug`ida kesmaning birinchi uchining koordinatasi  $(X_1, Y_1)$  ni ko`rsatmaslik ham mumkin.

$\langle N \rangle$ LINE- (X_2, Y_2) , $\langle P \rangle$

Bu holda kesmaning uchi sifatida oldingi koordinata olinadi. Uchlari (10,10), (20,80), (100,40) nuqtalarda bo`lgan uchburchakni quyidagicha ham chizish mumkin:

```
10 CLS:SCREEN 2
20 LINE (10,10)-(20,80)
30 LINE-(100,40)
40 LINE-(10,10)
50 GOTO 50
```

Buyruqdagi absolyut koordinatalar (X_1, Y_1) , (X_1, Y_2) o`rniga nisbiy koordinatalar ham yozish mumkin. Buning uchun nisbiy koordinatalardan oldin STEP so`zini ko`rsatish kerak;

$\langle N \rangle$ LINE STEP $(\Delta X_1, \Delta Y_1)$ -STEP $(\Delta X_2, \Delta Y_2)$, $\langle P \rangle$

yoki

$\langle N \rangle$ LINE STEP $(\Delta X_1, \Delta Y_1)$ -(X_2, Y_2), $\langle P \rangle$

yoki

$\langle N \rangle$ LINE (X_1, Y_1) -STEP $(\Delta X_2, \Delta Y_2)$, $\langle P \rangle$

yoki

$\langle N \rangle$ LINE-STEP $(\Delta X_2, \Delta Y_2)$, $\langle P \rangle$

Agar ekranda hosil qilingan chizmaning oxirgi nuqtasining koordinatasi (a, b) bo`lsa, u holda yuqoridagi buyruqlar yordamida hosil qilingan kesma uchlarining koordinatasi (STEP ko`rsatilganlarida) mos ravishda $(a + \Delta X_1, b + \Delta Y_1)$ va $(a + \Delta X_2, b + \Delta Y_2)$ bo`ladi.

**b) To`g`ri to`rtburchakni chizish.** To`rtburchakni chizish uchun

$\langle N \rangle$ LINE (X_1, Y_1) -(X_2, Y_2), $\langle P \rangle$, B

buyrug`idan foydalanish mumkin. Bu buyruqda  $(X_1, Y_1)$  va  $(X_2, Y_2)$ -to`g`ri to`rtburchak mos uchlarining koordinatasi. Agar buyruqda $\langle P \rangle$ rang ko`rsatilmasa, uning yo`qligini ko`rsatuvchi vergul belgisi turishi shart.

Buyruqdagi V harfi inglizcha to`g`ri to`rtburchak so`zining (BOX) birinchi harfidan olingan.

Quyida ichma-ich joylashgan kvadratlarni hosil qilish dasturi keltirilgan:

```
10 CLS:SCREEN 2
20 FOR I=0 TO 60 STEP 10
30 LINE (I+10,I+10)-(150-I, 150-I),,B
40 NEXT I
50 GOTO 50
```

v) **Bo`yalgan to`g`ri to`rtburchakni chizish.** Bo`yalgan to`g`ri to`rtburchakni chizish uchun quyidagi buyruqdan foydalaniladi:

$\langle N \rangle \text{ LINE } (X_1, Y_1)-(X_2, Y_2), \langle P \rangle, BF$

Bu buyruq ham to`g`ri to`rtburchak chizish buyrug`idek bo`lib, undan farqi shundaki, hosil bo`lgan to`g`ri to`rtburchak  $\langle P \rangle$  rangga bo`yaladi. Misol:

```
10 REM ZINAPOYA
20 CLS: SCREEN 2: PSET(10,10)
30 FOR I=1 TO 7
40 LINE-STEP(30,20),,BF
50 NEXT
60 GOTO 60
```


Aylana, ellips va ularni yoylari chizish.

Aylana, ellips va ularning yoylarini chizish uchun maxsus buyruqlar mavjuddir.

Aylana chizish uchun uning markazini, radiusini va aylana rangini tanlash lozim bo`ladi. Bu buyruqni ko`rinishi quyidagicha:

$\langle N \rangle \text{ CIRCLE}(X, Y), R, \langle P \rangle$

Bu yerda (X, Y) - aylana markazining koordinatasi; R -radius va $\langle R \rangle$ -rang.




Quyidagi dastur markazi (100,100) nuqtada bo`lgan har xil rangli ichma-ich joylashgan aylanalarni hosil qiladi. Rang ekran rangida bo`lgan aylana ko`rinmaydi.

```

10 CLS: SCREEN 2
20 FOR I=0 TO 15
30 CIRCLE(100,100),6*I+4,I
40 NEXT
50 GOTO 50

```

Aylananing yoyini chizish uchun, kerak bo`ladigan parametrlardan tashqari yoy uchlarning aylana markazidan o`tuvchi gorizontaal o`q bilan hosil qilgan burchaklarini ham berish kerak bo`ladi. Bu burchaklar $0 \leq \alpha \leq \beta \leq 2\pi$ oraliqda o`zgaradi va radianda beriladi.



Ekranda aylana yoyini hosil qilish buyrug`ining ko`rinishi quyidagicha bo`ladi:

<N> CIRCLE(X,Y),R,<P>, α , β

Misol,

```

10 PI=4*ATN(1)
20 CLS: SCREEN 2
30 FOR I=1 TO 15
40 CIRCLE(50,120),6*I+4,I,PI/6,PI/3
50 NEXT
60 GOTO 60

```

Ma`lumki, aylana sektorini hosil qilish uchun aylana yoyi uchlarning aylana markazi bilan radius orqali tutashtirish kifoya. Kompyuter ekranida buni hosil qilish uchun yuqorida buyruqda α va β ning qiymatlarini manfiy qilib olish yetarli.

<N> CIRCLE(X,Y),R,<P>,- α ,- β

Bunda burchaklar sifatida ularning alfavit qiymatlari olinib, manfiy ishora tutashtirish lozimligini bildiradi. Shuni ta`kidlash lozimki, burchaklarning biri manfiy, biri esa musbat bo`lishi ham mumkin. Qaysi bir burchak manfiy bo`lsa, shu yoy uchi markaz bilan tutashtiriladi.

Yuqoridagi dasturda 40-satni

40 CIRCLE(50,120),6*I+4,I,-PI/6,-PI/3

ko`nishda o`zgartirib, aylana sektorini hosil qilishimiz mumkin.

Ellips, ellips sektori va ellips yoyi chizish uchun yuqoridagi bandlarda keltirilgan barcha mulohazalar o`rinli bo`lib, faqatgina buyruqda bir parametr qo`shiladi.

<N> CIRCLE(X,Y),R,<P>, α , β ,M

Bu yerda **M** parametr o`qlar nisbatini bildiradi. Uning qiymati

$$1/260 \leq M \leq 260$$

oraliqda o`zgaradi.

Agar $M < 1$ bo`lsa, ellips Y o`qi bo`yicha siqiladi, agarda  $M > 1$  bo`lsa, X o`qi bo`yicha siqiladi. Shuni ta`kidlash lozimki, bu buyruq yordamida o`qlari mos koordinata o`qlariga parallel bo`lgan ellipsni chizish mumkin. Yuqorida keltirilgan buyruq eng to`liq ko`rinishda bo`lib, undagi biror parametr (<P>, $\alpha$ , $\beta$ ,M) ko`rsatilmasa uning yo`qligini ko`rsatuvchi vergul bo`lishi shart. Masalan ellips uchun buyruqni

<N> CIRCLE(X,Y),R,,,M

ko`rinishda ham berish mumkin.

Misollar,

Silindrni chizish dasturi.

```
10 CLS: SCREEN 2
20 LINE(40,40)-(40,150)
30 LINE(100,40)-(100,150)
40 CIRCLE(70,40),30,,,1/4
50 CIRCLE(70,150),30,,,1/4
60 K$=INKEY$: IF K$=" " THEN 60
```

Shar:

```
10 CLS: SCREEN 2
20 CIRCLE(128,96),80
30 CIRCLE(128,96),80,,,1/4
40 CIRCLE(128,96),80,,,4
50 K$=INKEY$: IF K$=" " THEN 50
```

Markazi bir nuqtada bo`lgan ellipslar:

```
10 CLS: SCREEN 2:COLOR 15,1,1
20 FOR I=24 TO 80 STEP 4
30 CIRCLE(128,96),96,,,10/I
40 CIRCLE(128,96),96,,,I/10
```

50 K\$=INKEY\$: IF K\$="'" THEN 50

Quyidagi dastur go`yoki konusni biror ellips bo`yicha bukish tasvirini hosil qiladi:

10 CLS: SCREEN 2

20 FOR I=1 TO 125

30 CIRCLE((110+100*COS(I/20), 90+80*SIN(I/20)),45-I/3,1

40 NEXT

50 K\$=INKEY\$: IF K\$="'" THEN 50

Nazariy savollar.

1. Kesmalar chizish buyrug`i ishtirokida misollar ko`rsating.
2. To`g`ri to`rtburchak chizish buyruqlarining turlarini sanab o`ting.
3. Uy tasvirini chizuvchi dastur tuzing.
4. Ichma-ich chizilgan to`rtburchaklarni hosil qiluvchi dastur tuzing.
5. Aylana, ellips va ularning yoylari qanday chiziladi?
6. Ellips sektori va shar qanday chiziladi?
7. Mashinaning tasvirini ekranda chiqaruvchi dastur tuzing.
8. Aylana ichida aylanalar chizuvchi dastur tuzing.

Muammoli savollar

1. 10 CLS: SCREEN2

20 LINE(101,10)-(101,30)

30 LINE(35,98)-(75,98)

40 GOTO 40

dastur natijasida ekranda nima hosil bo`ladi.

2. Ichma-ich joylashgan ikki xil rangdagi to`g`ri to`rtburchaklarni tasvirlash dasturini tuzing.
 3. ILM – so`zini yozish dasturini yozing.
 4. To`g`ri to`rtburchakni chizishda B – harfi nimani bildiradi.
 5. Ichma-ich joylashgan har xil rangdagi aylanalarni ekranga chiqarish dasturi tuzilsin.
 6. 10 SLS: SCREEN2
- 20 LINE(40,40)-(40)
- 30 LINE(100)-(100,50)
- 40 CIRSLE(70,40),30 ..., 1/4
- 50 CIRSLE(5),10
- 60 GOTO 60
- a. dastur xatosini aniqlang.

7. Tiko avtomobilini tasvirini ekranga chiqaruvchi dastur tuzing.
8. Aylana sektorini hosil qiluvchi dasturni tuzing.

**8-Ma'ruza. Bir paytda kompyuter ekranida grafik va matnni tasvirlash.
Geometrik shakllarni va ularning parametrlarini tasvirlash
dasturlarini tuzish.**

Reja:

1. Bir paytda kompyuter ekranida grafik va matnni tasvirlash.
2. DRAW buyrug'i.
3. DRAW buyrug'i ishtirokida dasturlar tuzish.
4. Shakllarni harakatlantirish.
5. Kompyuterda fazobiy shakllarni chizish.
6. Ekranida shakllarni harakatlantirish programmalarini tuzish.

Kompyuter ekranida hosil bo'ladigan tasvir grafik ekranda yoki matnli ekranda olinadi. Matnli ekran bilan ishlash jarayonida ma'lumotlarni ishlash jarayonida ma'lumotlarni chiqarishda PRINT yoki PRINT USING buyruqidan foydalanilgan. Grafik holatda ishlash jarayonida to'g'ridan-to'g'ri bu buyruqlar bilan ishlash ma'lumotlarning buzilishiga olib keladi. Chunki bu buyruqlar ekranni matnli holatga o'tkazadi. Ikkala ekrandagi ma'lumotlarni bir paytda kompyuter ekranida hosil qilish uchun fayllardan foydalaniladi. Buning uchun faylni ochish lozim va faylga ma'lumotlarni yozish uchun PRINT# buyrug'idan foydalanish yetarli. Fayl sifatida grafik ekran olinib, uning nomi uchun grafik ekrannig shartli nomi "GRP:" olinadi.

Chiqarilayotgan belgining o'rnini aniqlashda PSET yoki PRESET buyruqlaridan foydalaniladi. Bu buyruqlardagi nuqtaning koordinatasi chiqarilayotgan matndagi birinchi belgining chap yuqori burchagining koordinatasiga mos keladi. Agar belgining chiqarish o'rnini ko'rsatilmagan bo'lsa, u holda chiqarish oxirgi to'xtatilgan joydan davom ettiriladi. Xuddi shuningdek, belgi quyi satrga joylashmasa, u yuqori satrda hosil bo'ladi.

Quyidagi dastur kompyuter ekranining yarmida har xil rangdagi yozuvlar va ekranning o'ng tomonida doira, kvadrat va uchburchak chizadi va yozib qo'yadi:

```
10 COLOR 15,15,15: SCREEN 2
20 OPEN "GRP:" AS #1
30 LINE (0,0)-(255,191),1,B
40 LINE (120,0)-(120,191),1
50 PRESET (8,32)
60 FOR I=1 TO 14: COLOR I
70 PRINT#1, "SALOM"
80 PRESET STEP (8,0): NEXT I
90 PRESET (164,16): COLOR I
100 PRINT#1, "DOIRA"
110 CIRCLE (188,64),30:PAINT(188,64)
```

```

120 PRESET(144,112): COLOR 4
130 PRINT #1, "KVADRAT"
140 LINE (136,128)-STEP(40,40),8,BF
150 PRESET (186,128): COLOR 4
160 PRINT #1, "UCHBURCHAK"
170 LINE (208,128)-STEP(30,30)
180 LINE STEP (30,0): LINE STEP (0,-30):
190 K$=INKEY$: IF K$=" " THEN END ELSE 190

```

Grafik holatda ishlash jarayonida dasturda INPUT buyrug'i uchrab qolsa, bu buyruq avtomatik tarzda grafik holatdan matnli holatga o'tilishiga sabab bo'ladi. Chunki bu buyruq bajarilganda ekranda "?" belgisi hosil bo'ladi. Grafik holatni buzmasdan ma'lumotlarni kiritish uchun

INKEY\$, LINE INPUT yoki INPUT\$

buyruqlaridan foydalaniladi.

DRAW buyrug'i


Grafik holatda ishlash jarayonida ixtiyoriy shakllarni kompyuter ekranida hosil qilishning qulay usullaridan biri DRAW (chizish) buyrug'idir.

DRAW buyrug'ining umumiy ko'rinishi quyidagicha:

<N> DRAW "GML buyruq[lari]"

GML tilining buyruqlari quyidagi jadvalda keltirilgan:

GML tilining buyruqlari

№	Buyruq	Mazmuni
1	Mn	M-yo`nalish bo`yicha qalamning harakati $M = \{U, E, R, F, D, G, L, H\}$ Yo`nalishlar orasidagi burchak <math>45^\circ</math></p>  <p>n- butun son bo`lib, yo`nalish bo`yicha

№	Buyruq	Mazmuni
		necha birlik siljish kerakligini ko'rsatadi. Birlik sifatida bitta nuqta olinadi.
2	$M_{x,y}$	Qalamning turgan o'rnidan koordinatasi (x,y) bo'lgan nuqtaga siljitish.
3	$M_{\pm x, \pm y}$	Qalamning turgan o'rnidan absissa o'qi bo'yicha x birlikka, ordinata o'qi bo'yicha y birlikka siljitish (ishora ko'rsatilishi shart)
4	C_p	Chizilayotgan kesmaning rangi ($p=0,1,...,15$) rangning kodi. Bu buyruq boshqa rang tanlanmaguncha ta'sir etadi.
5	S_m	Masshtab ko'paytuvchi m ni berish ($1 \leq m \leq 255$). Bu buyruq bajarilganda barcha harakat buyruqlarining parametrlari $m/4$ marta oshadi.
6	A_t	Shaklni ($90 \cdot t$) gradusga soat mili yo'nalishiga teskari yo'nalishda burish. Burish shaklini chizish boshlangan nuqtaga nisbatan amalga oshiriladi. Boshqa A_t tanlanmaguncha bu buyruq barcha buyruqlarga ta'sir etadi.
7	$X;$	GML buyruqlarini satrli o'zgaruvchi yordamida tasvirlash. Satr nomi oldiga X harfi va nom oxiriga (;) qo'yilishi shart.

GML tilining buyruqlari asosan rangli qalamni ekranda bir nuqtadan boshqa nuqtaga siljitishdan iborat. Bunda 3 ta hol bo'lishi mumkin:

Chizib siljish:

$H20; U30; M+15, -20; M100,30.$

Chizmasdan siljish. Bu holda GML buyruqlari oldiga V harfi qo'yib yoziladi:

$BH20; BU30; BM+15, -20; BM100,30.$

Chizib siljish va chizish boshlangan nuqtaga qaytish. Bu holda buyruqlar oldiga N harfi qo'yib yoziladi:

$NH20; NU30; NM+15, -20; NM100,30.$

Biror kesmani chizish uchun kesmaning boshlang'ich nuqtasini yo'nalishini va uzunligini berish lozim.

Boshlang'ich nuqtani berishda:

1. $BM_{x,y}$ yoki $BM_{\pm x, \pm y}$ dan;
2. M_n dan;
3. PSET yoki PRESET dan;

Kesma yo'nalishini berish uchun:

1. $M_{x,y}$ yoki $M_{\pm x, \pm y}$ dan;
2. M_n dan buyruqlarini birortasidan foydalanish mumkin.

Ma'lumki DRAW buyrug'i yordamida ixtiyoriy B burchakli kesmani chizish qiyin. Ammo buni quyidagicha amalga oshirish mumkin:

100 DRAW "M ± 1 *COS(B); ± 1 *COS(B);"

Misol:

1. To'rtburchak:

```
10 SCREEN 2
20 DRAW "BM80,50 R100 G50 L100 E50"
```

2. Ixtiyoriy ko'pburchak:

```
10 SCREEN 2
20 DRAW "BM20,20 M60,30 M100,10 M120,120 M30,60 M20,20"
```

3. Broun harakati:

```
10 SCREEN 2
20 PSET (128,96)
30 X=INT(RND(1)*256): Y=INT(RND(1)*192);
40 DRAW "M=X;=Y;"
```

4. Kattalashib boruvchi to'g'ri to'rtburchaklar:

```
10 SCREEN 2
20 PSET(10,10)
30 FOR M=4 TO 80 STEP 4
40 DRAW "S=M; D4 R6 U4 L6"
50 NEXT
```

Shakllarni harakatlantirish. Grafik shakllarni ekranda tasbirlash va ularni harakatlantirish. Ekranda shakllarni harakatlantirish programmalarini tuzish.

Harakatlanuvchi shakllarni kompyuter ekranida hosil qilishning ikki yo`li mavjud. Birinchidan, harakatlanuvchi shaklni tashkil qiluvchi elementlar nuqta, kesma, siniq chiziq va h.k. kompyuter ekranida ma`lum rangda hosil qilinsa, ikkinchidan, bu shakl ekranning rangida yana bir marta hosil qilinadi, go`yoki o`chiriladi. Shu zaylda shaklning hosil qilish o`rnini o`zgartirib turish harakatlanayotgan tasavvurni hosil qiladi. Buni misollarda ko`rib o`tamiz.

1. **Matnli holat.** Matnli holatda ixtiyoriy belgini, harfni, so`zlarni va h.k. ekranning turli tomonlariga harakatlantirish mumkin. Masalan,

```
10 CLS: 'EKRANNI TOZALASH
20 FOR K=1 TO 32
30 LOCATE K, 5: ? SPC(5)
40 LOCATE K, 5: ? "SALOM"
50 NEXT K
```

"SALOM" so`zini chapga harakatlantirish uchun yuqoridagi dasturni qo`yidagicha o`zgartirish lozim:

```
30 LOCATE 32-K, 5: ? SPC(5)
40 LOCATE 32-K, 5: ? "SALOM"
```

So`zni ekranning pastiga harakatlantirish uchun yuqoridagi dasturga:

```
20 FOR K=1 TO 22
30 LOCATE 5, K: ? SPC(5)
40 LOCATE 5, K: ? "SALOM"
50 NEXT
```

o`zgartirish kiritish yetarli.

Ayrim hollarda harakat juda tezlashib ekranda belgini anglash qiyin bo`lishi mumkin. Bu hollarda harakat tezligini tartibga solib turish uchun kerakli joyda

<N> FOR I=1 TO M: NEXT I

buyrug`ini kiritish mumkin. M ning qiymatini o`zgartirib turish hisobiga harakatni tezlatish yoki sekinlatish mumkin.

2. **Grafik holat.** Bu holatda ham shaklni tashkil qiluvchi elementlarni (nuqta, kesma, siniq chiziq va h.k.) bir chizib, bir o`chirib shaklni harakatlantirish mumkin. Misol:

```

10 REM "HARAKATLANUVCHI NUQTA"
20 NEW
30 SCREEN 2
40 FOR K=1 TO 200
50 PSET(K, 60)
60 NEXT K
70 END

```

Kesmani o`ngga harakatlantirish uchun dasturni quyidagicha o`zgartirish yetarli:

```

55 PRESET (K-10, 96)
65 GOTO 40

```

Pastga harakatlantirish uchun

```

50 PSET (125, K)
55 PRESET (125, K-10)

```

Quyidagi dastur aylanani ekran bo`yicha o`ngga va chapga sirpanishini ta`minlaydi. Ammo aylanani sirpanishi juda ham tabiiy bo`la olmaydi, chunki aylanani hosil qilish va o`chirish silliq kechmaydi.

```

10 COLOR 15, 4, 4: SCREEN 2
20 X=5: SX=5
30 CIRCLE (X, 96),10,15
40 CIRCLE (X, 96),10, 4
50 X=X+ SX
60 IF X<5 OR X>250 THEN SX=-SX
70 CIRCLE (X, 96), 10,15
80 GOTO 40

```

Kompyuterda fazoviy shakllarni chizish

Fazoviy shakllarni (silindr, piramida va h.k.) yuqorida keltirilgan buyruqlar yordamida kompyuter ekranida qog`ozdagidek ko`rinishini hosil qilish mumkin.

Ayrim sodda geometrik shakllarni maxsus dasturlar yordamida chizish mumkin. Quyida bunga bir qancha misollar keltirilgan.

Piramida dasturi:

```

10 REM "PIRAMIDA"
20 INPUT "N NI KIRITING=";N
30 SCREEN 5: CLS: D=6.28/ N
40 X=130: Y=25: X2=X: Y2=Y
50 FOR I=0 TO 6.29 STEP D
60 X1=130+ 50*SIN(I): Y1=52-25*COS(I)

```

```

70 PRESET(X,Y): LINE-(X1,Y1)
80 LINE-(X2, Y2)
90 X2=X1: Y2=Y1: NEXT
100 K$=INKEY$: IF K$="" THEN 100
110 END

```

Konus dasturi:

```

10 REM KONUS
20 SCREEN 5: CLS
30 FOR I=0 TO 72 STEP 4
40 CIRCLE(120, 20+2*I), , 3.14, 6.28, 1 / 2
50 NEXT I
60 LINE(120,20)-(48,164)
70 LINE(120, 20)-(192,164)
80 K$=INKEY$: IF K$="" THEN 80

```

Sinusoida grafigi dasturi:

```

10 REM CINUSOIDA X=SIN(Z), Y=0 TEKISLIK
20 PI=4*ATN(1): X0=80: Y0=150
30 CLS : SCREEN 5
40 DRAW "C15 BM80, 150 NR100 NU120 NM87,40"
50 FOR F=0 TO 6*PI STEP PI / 50
60 Z=6*F: X=40*SIN(F)
70 PSET(X0+X+0,707*Z, Y0-0,707*Z)
80 NEXT F
90 K$=INKEY$: IF K$="" THEN 90

```

Ellips dasturi

```

10 REM ELLIPS
20 SLS: SCREEN 5
30 DRAW "BM80,150 NR100 NU120 NM180,40"
40 PI=4*ATN(1): Y=30
50 FOR F=0 TO 2*PI STEP PI / 50
60 Z=40*SIN(F): X=20*COS(F)
70 PSET(80+X+0,707*Z, 150-Y-0,707*Z)
80 NEXT F
90 K$=INKEY$: IF K$="" THEN 90

```

Nazariy savollar.

1. Bir paytning o`zida ekranda qanday qilib grafik va matnni tasvirlash mumkin?

2. DRAW buyrug`ining formatini ko`rsatib bering.
3. Ekranda romb va konuslarning tasvirini chizib bering.
4. DRAW buyrug`i ishtirokida uyning tasvirini chizuvchi dastur tuzing.
5. Kompyuterda shakllarni qanday qilib xarakatlantiriladi?
6. Kompyuterda fazoviy shakllarni chizishga misollar ko`rsating.
7. Kubni tasvirlovchi dastur tuzing.
8. Xarakatlanuvchi nuqta dasturini tuzing.

Muammoli savollar

1. Grafik holatni buzmasdan ma`lumotlarni kiritish uchun INPUT buyrug`i qanday ishlatiladi?
2. $M=\{U, E, R, F, D, G, L, H\}$ – bular sizga nimani eslatadi?
3. 10 SCREEN2
4. 20 PSET(20,20)
5. 30 FOR M=14 TO 80 STEP 4
6. 40 DRAW "S=M; D8; R8; U8; L8"
7. NEXT
8. Konus shaklini tasvirlash dasturini tuzing.
9. Shakllarni xarakatlantirishga dastur tuzing.
10. Kompyuterda fazoviy shakllar chizing.
11. Kubni o`z o`qi atrofida aylanish dasturini tuzing.
12. Biror bir o`yinning dasturini tuzing.

FOYDALANILGAN ADABIYOTLAR

1. Karimov I. A. O‘zbekiston buyuk kelajak sari.—Toshkent.: «O‘zbekiston», 1998.—528 b.
2. Barkamol avlod — O‘zbekiston taraqqiyotining poydevori.(O‘zbekiston Respublikasining «Ta‘lim to‘g‘risida» va «Kadrlar tayyorlash milliy dasturi to‘g‘risida»gi qonunlar).—T.: «Shark», 1998.—64 b.
3. Informatika: Kasb-xunar kollejlari uchun o‘quv dasturi.Mualliflar jamoasi: A.A.Abduqodirov, R. D. Aloev, R. R. Boqiev va boshqalar—T.:2000.—12 b.
4. Raxmonqulova S. I. IBM PS shaxsiy kompyuterida ishlash.—Toshkent, 1998. —224 b.
5. U. Yuldashev, R. R. Boqiev, M. E. Mamarajabov. EXCEL 97: O‘quv qo‘llanma.—T., 2000.—40 b.
6. Yuldashev, M. E. Mamarajabov, K. A. Mirvalieva. POWER POINT 97: O‘quv qo‘llanma.—T., 2001.—32 b.
7. U. Yuldashev, Sh. K. Raxmatullaeva. Microsoft WINDOWS: O‘quv qo‘llanma.—T., 2001.—29 b.
8. Yuldashev, Sh. K. Raxmatullaeva. Microsoft WORD 97: O‘quv qo‘llanma.—T., 2001.—47 b.
9. O‘zbekiston Davlat ta‘lim standarti: O‘rta maxsus, kasb-xunar ta‘limi umumta‘lim fanlari. — «Ma‘rifat», №86, 2000 y. 4 noyabr.
10. G‘ulomov S. S., Shermuxamedov A. T., Begalov B. A. “Iktisodiy informatika” : Darslik Akademik S. S. G‘ulomovning umumiy taxriri ostida.—T.: «O‘zbekiston», 1999.—528 b.
11. G‘ulomov S. S. va boshqalar. Axborot tizimlari va texnologiyalari: Oliy o‘quv yurti talabalari uchun darslik Akademik S. S. G‘ulomovning umumiy taxriri ostida.—T.: «Shark», 2000.—592 b
12. S.S.G‘ulomov va boshqalar. «Axborot tizimlari va texnologiyalari». – Toshkent. «Sharq», 2000 yil.
13. V.L. Broydo. «Ofis texnikasi: Boshqarish va ish yuritish uchun». - Toshkent. «Mehnat», 2001 yil.
14. A.Sattorov, B.Qurbanboev. «Informatika va hisoblash texnikasi asoslari». – Toshkent. «O‘qituvchi», 1996 yil.

[illegible]

Mundarija

Soʻz boshi.....	3
1-Maʼruza. Takrorlash operatori va ularning qoʻllanilishi. Takrorlanuvchi algoritmnlarni dasturlash. Takrorlanuvchi jarayonlarni hisoblash algoritmnlari va dasturlarini tuzish.....	4
2-Maʼruza. Jadval kattaliklar. Massiv. Massivlarni ifodalash va ular ustida amallar bajarish. Jadval kattaliklarini qayta ishlash.....	8
3-Maʼruza. Foydalanuvchi funksiyasi. Qism dastur. Foydalanuvchi funksiyalar va qism dasturlarini tashkil qilish. Funksiya va qism dasturlarga murojaat etish.....	12
4-Maʼruza. Satr kattaliklarni qayta ishlash. Satr kattaliklarni qayta ishlovchi funksiyalar. Satr kattaliklarni qayta ishlovchi dasturlar tuzish.....	16
5-Maʼruza. Beysik (Paskal) tilining imkoniyatlari. Dasturni xotiraga yozish va undan oʻqish. Beysik (Paskal) tilining buyruqlari va ularning qoʻllanilishi.....	23
6-Maʼruza. Beysik (Paskal) tilining grafik imkoniyatlari. Funksiya grafigini belgilar yordamida tasvirlash. Funksiyani xisoblash va grafigini ekranda tasvirlash dasturlarini tuzish.....	30
7-Maʼruza. Ekranda turli grafik shakllarni hosil qilish. Kesma, toʻgʻri toʻrtburchaklar chizish. Ekranda turli grafik shakllarni hosil qilish uchun dasturlar tuzish. Aylana, ellips va ularni yoylari chizish.....	34
8-Maʼruza. Bir paytda kompyuter ekranida grafik va matnni tasvirlash. Geometrik shakllarni va ularning parametrlarini tasvirlash dasturlarini tuzish.....	40
Foydalanilgan adabiyotlar.....	48

